

# SWI-Prolog SGML/XML parser

*Version 1.0.15, September 2002*

*Jan Wielemaker*

SWI,  
University of Amsterdam  
The Netherlands  
**E-mail:** `jan@swi.psy.uva.nl`

## **Abstract**

Markup languages are an increasingly important method for data-representation and exchange. This article documents the package **sgml2pl**, a foreign library for SWI-Prolog to parse SGML and XML documents, returning information on both the document and the document's DTD. The parser is designed to be small, fast and flexible.

# Contents

<b>1</b>	<b>Introduction</b>	<b>2</b>
<b>2</b>	<b>Bluffer's Guide</b>	<b>2</b>
2.1	'Goodies' Predicates . . . . .	3
<b>3</b>	<b>Predicate Reference</b>	<b>4</b>
3.1	Loading Structured Documents . . . . .	4
3.2	Handling white-space . . . . .	6
3.3	XML documents . . . . .	7
3.3.1	XML Namespaces . . . . .	7
3.4	DTD-Handling . . . . .	8
3.4.1	The DOCTYPE declaration . . . . .	11
3.5	Extracting a DTD . . . . .	11
3.6	Parsing Primitives . . . . .	11
3.6.1	Partial Parsing . . . . .	16
<b>4</b>	<b>Processing Indexed Files</b>	<b>17</b>
<b>5</b>	<b>External entities</b>	<b>18</b>
<b>6</b>	<b>Writing markup</b>	<b>19</b>
<b>7</b>	<b>Unsupported features</b>	<b>19</b>
<b>8</b>	<b>Installation</b>	<b>20</b>
8.1	Unix systems . . . . .	20
<b>9</b>	<b>Acknowledgements</b>	<b>21</b>
<b>A</b>	<b>Summary of Predicates</b>	<b>22</b>

## 1 Introduction

Markup languages have recently regained popularity for two reasons. One is document exchange, which is largely based on HTML, an instance of SGML, and the other is for data exchange between programs, which is often based on XML, which can be considered a simplified and rationalised version of SGML.

James Clark's SP parser is a flexible SGML and XML parser. Unfortunately it has some drawbacks. It is very big, not very fast, cannot work under event-driven input and is generally hard to program beyond the scope of the well designed generic interface. The generic interface however does not provide access to the DTD, does not allow for flexible handling of input or parsing the DTD independently of a document instance.

The parser described in this document is small (less than 50 kbytes executable on a Pentium or 80 kbytes on a SPARC), fast (between 2 and 5 times faster than SP), provides access to the DTD, and provides flexible input handling.

The document output is equal to the output produced by *xml2pl*, an SP interface to SWI-Prolog written by Anjo Anjewierden.

## 2 Bluffer's Guide

This package allows you to parse SGML, XML and HTML data into a Prolog data structure. The high-level interface defined in **sgml** provides access at the file-level, while the low-level interface defined in the foreign module works with Prolog streams. Please use the source of **sgml.pl** as a starting point for dealing with data from other sources than files, such as SWI-Prolog resources, network-sockets, character strings, *etc.* The first example below loads an HTML file.

```
<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 3.2//EN">
```

```
<html>
<head>
<title>Demo</title>
</head>
<body>
```

```
<h1 align=center>This is a demo</title>
```

```
<p>Paragraphs in HTML need not be closed.
```

```
<p>This is called 'omitted-tag' handling.
</body>
</html>
```

```
?- load_html_file('test.html', Term),
   pretty_print(Term).
```

```

[ element(html,
  [],
  [ element(head,
    [],
    [ element(title,
      [],
      [ 'Demo'
    ])
  ]),
  element(body,
    [],
    [ '\n',
      element(h1,
        [ align = center
        ],
        [ 'This is a demo'
        ]),
      '\n\n',
      element(p,
        [],
        [ 'Paragraphs in HTML need not be closed.\n'
        ]),
      element(p,
        [],
        [ 'This is called 'omitted-tag\' handling.'
        ])
    ])
  ])
].

```

The document is represented as a list, each element being an atom to represent CDATA or a term `element(Name, Attributes, Content)`. Entities (e.g. `&lt;`;) are returned as part of CDATA, unless they cannot be represented. See [load\\_sgml\\_file/2](#) for details.

## 2.1 ‘Goodies’ Predicates

These predicates are for basic use of the library, converting entire and self-contained files in SGML, HTML, or XML into a structured term. They are based on [load\\_structure/3](#).

**load\_sgml\_file(+File, -ListOfContent)**

Same as `load_structure(File, ListOfContent, [dialect(sgml)])`.

**load\_xml\_file(+File, -ListOfContent)**

Same as `load_structure(File, ListOfContent, [dialect(xml)])`.

### **load\_html\_file(+File, -Content)**

Load *File* and parse as HTML. Implemented as below. Note that **load\_html\_file/2** reuses a cached DTD object as defined by **dtd/2**. As DTD objects may be corrupted while loading erroneous documents sharing is undesirable if the documents are not known to be correct. See **dtd/2** for details.

```
load_html_file(File, Term) :-
    dtd(html, DTD),
    load_structure(File, Term,
        [ dtd(DTD),
          dialect(sgml),
          shorttag(false)
        ]).
```

## **3 Predicate Reference**

### **3.1 Loading Structured Documents**

SGML or XML files are loaded through the common predicate **load\_structure/3**. This is a predicate with many options. For simplicity a number of commonly used shorthands are provided: **load\_sgml\_file/2**, **load\_xml\_file/2**, and **load\_html\_file/2**.

#### **load\_structure(+Source, -ListOfContent, +Options)**

Parse *Source* and return the resulting structure in *ListOfContent*. *Source* is either a term of the format **stream(StreamHandle)** or a file-name. *Options* is a list of options controlling the conversion process.

A proper XML document contains only a single toplevel element whose name matches the document type. Nevertheless, a list is returned for consistency with the representation of element content. The *ListOfContent* consists of the following types:

#### ***Atom***

Atoms are used to represent CDATA. Note this is possible in SWI-Prolog, as there is no length-limit on atoms and atom garbage collection is provided.

#### **element(Name, ListAttributes, ListOfContent)**

*Name* is the name of the element. Using SGML, which is case-insensitive, all element names are returned as lowercase atoms.

*ListOfAttributes* is a list of *Name=Value* pairs for attributes. Attributes of type CDATA are returned literal. Multi-valued attributes (**NAMES**, *etc.*) are returned as a list of atoms. Handling attributes of the types **NUMBER** and **NUMBERS** depends on the setting of the **number(+NumberMode)** attribute through **set\_sgml\_parser/2** or **load\_structure/3**. By default they are returned as atoms, but automatic conversion to Prolog integers is supported. *ListOfContent* defines the content for the element.

#### **entity(Code)**

If a character-entity (e.g. **&#913;**) is encountered that cannot be represented in

the Prolog character set, this term is returned, representing the referred character code.

**entity**(*Name*)

If an entity refers to a character-entity holding a single character, but this character cannot be represented in the Prolog character set, this term is returned. For example, the HTML input text `&Alpha; &lt; &Beta;` is returned as below. Please note that entity names are case sensitive in both SGML and XML.

```
[ entity('Alpha'), ' < ', entity('Beta') ]
```

This is a special case of **entity**(*Code*), intended to handle special symbols by their name rather than character code.

**sdata**(*Text*)

If an entity with declared content-type **SDATA** is encountered, this term is returned holding the data in *Text*.

**ndata**(*Text*)

If an entity with declared content-type **NDATA** is encountered, this term is returned holding the data in *Text*.

**pi**(*Text*)

If a processing instruction is encountered (`<?...?>`), *Text* holds the text of the processing instruction. Please note that the `<?xml ...?>` instruction is handled internally.

The *Options* list controls the conversion process. Currently defined options are:

**dtd**(*?DTD*)

Reference to a DTD object. If specified, the `<!DOCTYPE ...>` declaration is ignored and the document is parsed and validated against the provided DTD. If provided as a variable, the created DTD is returned. See section 3.5.

**dialect**(*+Dialect*)

Specify the parsing dialect. Supported are **sgml** (default), **xml** and **xmlns**. See section 3.3 for details on the differences.

**shorttag**(*+Bool*)

Define whether **SHORTTAG** abbreviation is accepted. The default is true for SGML mode and false for the XML modes. Without **SHORTTAG**, a `/` is accepted with warning as part of an unquoted attribute-value, though `/>` still closes the element-tag in XML mode. It may be set to false for parsing HTML documents to allow for unquoted URLs containing `/`.

**space**(*+SpaceMode*)

Sets the 'space-handling-mode' for the initial environment. This mode is inherited by the other environments, which can override the inherited value using the XML reserved attribute **xml:space**. See section 3.2.

**number**(*+NumberMode*)

Determines how attributes of type **NUMBER** and **NUMBERS** are handled. If **token** (default) they are passed as an atom. If **integer** the parser attempts to convert

the value to an integer. If successful, the attribute is passed as a Prolog integer. Otherwise it is still passed as an atom. Note that SGML defines a numeric attribute to be a sequence of digits. The `-` sign is not allowed and `1` is different from `01`. For this reason the default is to handle numeric attributes as tokens. If conversion to integer is enabled, negative values are silently accepted.

**defaults**(*+Bool*)

Determines how default and fixed values from the DTD are used. By default, defaults are included in the output if they do not appear in the source. If **false**, only the attributes occurring in the source are emitted.

**file**(*+Name*)

Sets the name of the file on which errors are reported. Sets the linenumber to 1.

**line**(*+Line*)

Sets the starting line-number for reporting errors.

**max\_errors**(*+Max*)

Sets the maximum number of errors. If this number is reached, an exception of the format below is raised. The default is 50.

```
error(limit_exceeded(max_errors, Max), _)
```

## 3.2 Handling white-space

SGPL2PL has four modes for handling white-space. The initial mode can be switched using the `space(SpaceMode)` option to `load_structure/3` and `set_sgml_parser/2`. In XML mode, the mode is further controlled by the `xml:space` attribute, which may be specified both in the DTD and in the document. The defined modes are:

**space**(*sgml*)

In SGML, newlines at the start and end of an element are removed.<sup>1</sup> This is the default mode for the SGML dialect.

**space**(*preserve*)

White space is passed literally to the application. This mode leaves all white space handling to the application. This is the default mode for the XML dialect.

**space**(*default*)

In addition to `sgml` space-mode, all consecutive white-space is reduced to a single space-character. This mode canonises all white space.

**space**(*remove*)

In addition to `default`, all leading and trailing white-space is removed from `CDATA` objects. If, as a result, the `CDATA` becomes empty, nothing is passed to the application. This mode is especially handy for processing ‘data-oriented’ documents, such as RDF. It is not suitable for normal text documents. Consider the HTML fragment below. When processed in this mode, the spaces between the three modified words are lost. This mode is not part of any standard; XML 1.0 allows only `default` and `preserve`.

---

<sup>1</sup>In addition, newlines at the end of lines containing only markup should be deleted. This is not yet implemented.

Consider adjacent `<b>bold</b>` `<ul>and</ul>` `<it>italic</it>` words.

### 3.3 XML documents

The parser can operate in two modes: `sgml` mode and `xml` mode, as defined by the `dialect(Dialect)` option. Regardless of this option, if the first line of the document reads as below, the parser is switched automatically into XML mode.

```
<?xml ... ?>
```

Currently switching to XML mode implies:

- *XML empty elements*  
The construct `<element [attribute...] />` is recognised as an empty element.
- *Predefined entities*  
The following entities are predefined: `lt (<)`, `gt (>)`, `amp (&)`, `apos (')` and `quot (")`.
- *Case sensitivity*  
In XML mode, names are treated case-sensitive, except for the DTD reserved names (i.e. `ELEMENT`, *etc.*).
- *Character classes*  
In XML mode, underscores (`_`) and colon (`:`) are allowed in names.
- *White-space handling*  
White space mode is set to `preserve`. In addition to setting white-space handling at the toplevel the XML reserved attribute `xml:space` is honoured. It may appear both in the document and the DTD. The `remove` extension is honoured as `xml:space` value. For example, the DTD statement below ensures that the `pre` element preserves space, regardless of the default processing mode.

```
<!ATTLIST pre xml:space nmtoken #fixed preserve>
```

#### 3.3.1 XML Namespaces

Using the *dialect* `xmlns`, the parser will interpret XML namespaces. In this case, the names of elements are returned as a term of the format

*URL:LocalName*

If an identifier has no namespace and there is no default namespace it is returned as a simple atom. If an identifier has a namespace but this namespace is undeclared, the namespace name rather than the related URL is returned.

Attributes declaring namespaces (`xmlns:ns=url`) are reported as if `xmlns` were not a defined resource.

In many cases, getting attribute-names as *url:name* is not desirable. Such terms are hard to unify and sometimes multiple URLs may be mapped to the same identifier. This may happen due to poor version management, poor standardisation or because the application doesn't care too much about versions. This package defines two call-backs that can be set using **set\_sgml\_parser/2** to deal with this problem.

The call-back `xmlns` is called as XML namespaces are noticed. It can be used to extend a canonical mapping for later use by the `urlns` call-back. The following illustrates this behaviour. Any namespace containing `rdf-syntax` in its URL or that is used as `rdf` namespace is canonised to `rdf`. This implies that any attribute and element name from the RDF namespace appears as `rdf:name`.

```
:- dynamic
    xmlns/3.

on_xmlns(rdf, URL, _Parser) :- !,
    asserta(xmlns(URL, rdf, _)).
on_xmlns(_, URL, _Parser) :-
    sub_atom(URL, _, _, _, 'rdf-syntax'), !,
    asserta(xmlns(URL, rdf, _)).

load_rdf_xml(File, Term) :-
    load_structure(File, Term,
        [ dialect(xmlns),
          call(xmlns, on_xmlns),
          call(urlns, xmlns)
        ]).
```

### 3.4 DTD-Handling

The DTD (**D**ocument **T**ype **D**efinition) is a separate entity in `sgml2pl`, that can be created, freed, defined and inspected. Like the parser itself, it is filled by opening it as a Prolog output stream and sending data to it. This section summarises the predicates for handling the DTD.

#### **new\_dtd**(+DocType, -DTD)

Creates an empty DTD for the named *DocType*. The returned DTD-reference is an opaque term that can be used in the other predicates of this package.

#### **free\_dtd**(+DTD)

Deallocate all resources associated to the DTD. Further use of *DTD* is invalid.

#### **load\_dtd**(+DTD, +File)

Define the DTD by loading the SGML-DTD file *File*. This predicate is defined using the low-level **open\_dtd/3** predicate:

```
load_dtd(DTD, DtdFile) :-
    open_dtd(DTD, [], DtdOut),
```

```

open(DtdFile, read, DtdIn),
copy_stream_data(DtdIn, DtdOut),
close(DtdIn),
close(DtdOut).

```

**open\_dtd**(+DTD, +Options, -OutputStream)

Open a DTD as an output stream. The option-list is currently empty. See **load\_dtd/2** for an example.

**dtd**(+DocType, -DTD)

Find the DTD representing the indicated *doctype*. This predicate uses a cache of DTD objects. If a doctype has no associated dtd, it searches for a file using the file search path **dtd** using the call:

```

...,
absolute_file_name(dtd(Type),
                  [ extensions([dtd]),
                    access(read)
                  ], DtdFile),
...

```

Note that DTD objects may be modified while processing erroneous documents. For example, loading an SGML document starting with `<?xml ...?>` switches the DTD to XML mode and encountering unknown elements adds these elements to the DTD object. Re-using a DTD object to parse multiple documents should be restricted to situations where the documents processed are known to be error-free.

**dtd\_property**(+DTD, ?Property)

This predicate is used to examine the content of a DTD. Property is one of:

**doctype**(DocType)

An atom representing the document-type defined by this DTD.

**elements**(ListOfElements)

A list of atoms representing the names of the elements in this DTD.

**element**(Name, Omit, Content)

The DTD contains an element with the given name. *Omit* is a term of the format `omit(OmitOpen, OmitClose)`, where both arguments are booleans (`true` or `false` representing whether the open- or close-tag may be omitted). *Content* is the content-model of the element represented as a Prolog term. This term takes the following form:

**empty**

The element has no content.

**cdata**

The element contains non-parsed character data. All data up to the matching end-tag is included in the data (*declared content*).

**rcdata**

As *cdata*, but entity-references are expanded.

**any**

The element may contain any number of any element from the DTD in any order.

**#pcdata**

The element contains parsed character data .

***element***

An element with this name.

**\*(*SubModel*)**

0 or more appearances.

**?(*SubModel*)**

0 or one appearance.

**+(*SubModel*)**

1 or more appearances.

**,(*SubModel1*, *SubModel2*)**

*SubModel1* followed by *SubModel2*.

**&(*SubModel1*, *SubModel2*)**

*SubModel1* and *SubModel2* in any order.

**|(*SubModel1*, *SubModel2*)**

*SubModel1* or *SubModel2*.

**attributes(*Element*, *ListOfAttributes*)**

*ListOfAttributes* is a list of atoms representing the attributes of the element *Element*.

**attribute(*Element*, *Attribute*, *Type*, *Default*)**

Query an element. *Type* is one of *cdata*, *entity*, *id*, *idref*, *name*, *nmtoken*, *notation*, *number* or *nutoken*. For DTD types that allow for a list, the notation *list*(*Type*) is used. Finally, the DTD construct (*a|b|...*) is mapped to the term *nameof(ListOfValues)*.

*Default* describes the sgml default. It is one of *required*, *current*, *conref* or *implied*. If a real default is present, it is one of *default*(*Value*) or *fixed*(*Value*).

**entities(*ListOfEntities*)**

*ListOfEntities* is a list of atoms representing the names of the defined entities.

**entity(*Name*, *Value*)**

*Name* is the name of an entity with given value. *Value* is one of

***Atom***

If the value is atomic, it represents the literal value of the entity.

**system(*Url*)**

*Url* is the URL of the system external entity.

**public(*Id*, *Url*)**

For external public entities, *Id* is the identifier. If an URL is provided this is returned in *Url*. Otherwise this argument is unbound.

**notations(*ListOfNotations*)**

Returns a list holding the names of all NOTATION declarations.

**notation**(*Name*, *Decl*)

Unify *Decl* with a list if `system(+File)` and/or `public(+PublicId)`.

### 3.4.1 The DOCTYPE declaration

As this parser allows for processing partial documents and process the DTD separately, the DOCTYPE declaration plays a special role.

If a document has no DOCTYPE declaration, the parser returns a list holding all elements and CDATA found. If the document has a DOCTYPE declaration, the parser will open the element defined in the DOCTYPE as soon as the first real data is encountered.

## 3.5 Extracting a DTD

Some documents have no DTD. One of the neat facilities of this library is that it builds a DTD while parsing a document with an *implicit* DTD. The resulting DTD contains all elements encountered in the document. For each element the content model is a disjunction of elements and possibly `#PCDATA` that can be repeated. Thus, if we found element *y* and CDATA in element *x*, the model is:

```
<!ELEMENT x - - (y|#PCDATA)*>
```

Any encountered attribute is added to the attribute list with the type `CDATA` and default `#IMPLIED`.

The example below extracts the elements used in an unknown XML document.

```
elements_in_xml_document(File, Elements) :-
    load_structure(File, _,
        [ dialect(xml),
          dtd(DTD)
        ]),
    dtd_property(DTD, elements(Elements)),
    free_dtd(DTD).
```

## 3.6 Parsing Primitives

**new\_sgml\_parser**(*-Parser*, *+Options*)

Creates a new parser. A parser can be used one or multiple times for parsing documents or parts thereof. It may be bound to a DTD or the DTD may be left implicit, in which case it is created from the document prologue or parsing is performed without a DTD. Options:

**dtd**(*?DTD*)

If specified with an initialised DTD, this DTD is used for parsing the document, regardless of the document prologue. If specified using as a variable, a reference

to the created DTD is returned. This DTD may be created from the document prologue or build implicitly from the document's content.

**free\_sgml\_parser**(*+Parser*)

Destroy all resources related to the parser. This does not destroy the DTD if the parser was created using the `dtd(DTD)` option.

**set\_sgml\_parser**(*+Parser, +Option*)

Sets attributes to the parser. Currently defined attributes:

**file**(*File*)

Sets the file for reporting errors and warnings. Sets the line to 1.

**line**(*Line*)

Sets the current line. Useful if the stream is not at the start of the (file) object for generating proper line-numbers.

**charpos**(*Offset*)

Sets the current character location. See also the `file(File)` option.

**dialect**(*Dialect*)

Set the markup dialect. Known dialects:

**sgml**

The default dialect is to process as SGML. This implies markup is case-insensitive and standard SGML abbreviation is allowed (abbreviated attributes and omitted tags).

**xml**

This dialect is selected automatically if the processing instruction `<?xml ...>` is encountered. See section 3.3 for details.

**xmlns**

Process file as XML file with namespace support. See section 3.3.1 for details. See also the `qualify_attributes` option below.

**qualify\_attributes**(*Boolean*)

How to handle unqualified attribute (i.e. without an explicit namespace) in XML namespace (`xmlns`) mode. Default and standard compliant is not to qualify such elements. If `true`, such attributes are qualified with the namespace of the element they appear in. This option is for backward compatibility as this is the behaviour of older versions. In addition, the namespace document suggests unqualified attributes are often interpreted in the namespace of their element.

**space**(*SpaceMode*)

Define the initial handling of white-space in PCDATA. This attribute is described in section 3.2.

**number**(*NumberMode*)

If `token` (default), attributes of type number are passed as a Prolog atom. If `integer`, such attributes are translated into Prolog integers. If the conversion fails (e.g. due to overflow) a warning is issued and the value is passed as an atom.

**encoding**(*Encoding*)

Set the initial encoding. The default initial encoding for XML documents is UTF-8

and for SGML documents ISO-8859-1. XML documents may change the encoding using the `encoding=` attribute in the header. Explicit use of this option is only required to parse non-conforming documents. Currently accepted values are `iso-8859-1` and `utf-8`.

**doctype**(*Element*)

Defines the toplevel element expected. If a `<!DOCTYPE` declaration has been parsed, the default is the defined doctype. The parser can be instructed to accept the first element encountered as the toplevel using `doctype()`. This feature is especially useful when parsing part of a document (see the `parse` option to `sgml_parse/2`).

**get\_sgml\_parser**(*+Parser, -Option*)

Retrieve information on the current status of the parser. Notably useful if the parser is used in the call-back mode. Currently defined options:

**file**(*-File*)

Current file-name. Note that this may be different from the provided file if an external entity is being loaded.

**line**(*-Line*)

Line-offset from where the parser started its processing in the file-object.

**charpos**(*-CharPos*)

Offset from where the parser started its processing in the file-object. See section 4.

**charpos**(*-Start, -End*)

Character offsets of the start and end of the source processed causing the current call-back. Used in **PceEmacs** to for colouring text in SGML and XML modes.

**source**(*-Stream*)

Prolog stream being processed. May be used in the `on_begin`, *etc.* callbacks from `sgml_parse/2`.

**dialect**(*-Dialect*)

Return the current dialect used by the parser (`sgml`, `xml` or `xmlns`).

**event\_class**(*-Class*)

The *event class* can be requested in call-back events. It denotes the cause of the event, providing useful information for syntax highlighting. Defined values are:

**explicit**

The code generating this event is explicitly present in the document.

**omitted**

The current event is caused by the insertion of an omitted tag. This may be a normal event in SGML mode or an error in XML mode.

**shorttag**

The current event (`begin` or `end`) is caused by an element written down using the *shorttag* notation (`<tag/value/>`).

**shortref**

The current event is caused by the expansion of a *shortref*. This allows for highlighting shortref strings in the source-text.

**doctype**(-*Element*)

Return the defined document-type (= toplevel element). See also **set\_sgml\_parser/2**.

**dtd**(-*DTD*)

Return the currently used DTD. See **dtd\_property/2** for obtaining information on the DTD such as element and attribute properties.

**context**(-*StackOfElements*)

Returns the stack of currently open elements as a list. The head of this list is the current element. This can be used to determine the context of, for example, CDATA events in call-back mode. The elements are passed as atoms. Currently no access to the attributes is provided.

**allowed**(-*Elements*)

Determines which elements may be inserted at the current location. This information is returned as a list of element-names. If character data is allowed in the current location, **#pcdata** is part of *Elements*. If no element is open, the *doctype* is returned.

This option is intended to support syntax-sensitive editors. Such an editor should load the DTD, find an appropriate starting point and then feed all data between the starting point and the caret into the parser. Next it can use this option to determine the elements allowed at this point. Below is a code fragment illustrating this use given a parser with loaded DTD, an input stream and a start-location.

```

...
seek(In, Start, bof, _),
set_sgml_parser(Parser, charpos(Start)),
set_sgml_parser(Parser, doctype(_)),
Len is Caret - Start,
sgml_parse(Parser,
  [ source(In),
    content_length(Len),
    parse(input) % do not complete document
  ]),
  get_sgml_parser(Parser, allowed(Allowed)),
...

```

**sgml\_parse**(+*Parser*, +*Options*)

Parse an XML file. The parser can operate in two input and two output modes. Output is either a structured term as described with **load\_structure/2** or call-backs on predefined events. The first is especially suitable for manipulating not-too-large documents, while the latter provides a primitive means for handling very large documents.

Input is a stream. A full description of the option-list is below.

**document**(+*Term*)

A variable that will be unified with a list describing the content of the document (see **load\_structure/2**).

**source**(*+Stream*)

An input stream that is read. This option *must* be given.

**content\_length**(*+Characters*)

Stop parsing after *Characters*. This option is useful to parse input embedded in *envelopes*, such as the HTTP protocol.

**parse**(*Unit*)

Defines how much of the input is parsed. This option is used to parse only parts of a file.

**file**

Default. Parse everything upto the end of the input.

**element**

The parser stops after reading the first element. Using **source**(*Stream*), this implies reading is stopped as soon as the element is complete, and another call may be issued on the same stream to read the next element.

**content**

The value **content** is like **element** but assumes the element has already been opened. It may be used in a call-back from **call**(**on\_begin**, *Pred*) to parse individual elements after validating their headers.

**declaration**

This may be used to stop the parser after reading the first declaration. This is especially useful to parse only the **doctype** declaration.

**input**

This option is intended to be used in conjunction with the **allowed**(*Elements*) option of **get\_sgml\_parser/2**. It disables the parser's default to complete the parse-tree by closing all open elements.

**max\_errors**(*+MaxErrors*)

Set the maximum number of errors. If this number is exceeded further writes to the stream will yield an I/O error exception. Printing of errors is suppressed after reaching this value. The default is 100.

**syntax\_errors**(*+ErrorMode*)

Defines how syntax errors are handled.

**quiet**

Suppress all messages.

**print**

Default. Pass messages to **print\_message/2**.

**style**

Print dubious input such as attempts for redefinitions in the DTD using **print\_message/2** with severity **informational**.

**call**(*+Event*, *:PredicateName*)

Issue call-backs on the specified events. *PredicateName* is the name of the predicate to call on this event, possibly prefixed with a module identifier. The defined events are:

**begin**

An open-tag has been parsed. The named handler is called with three arguments: *Handler*(*+Tag*, *+Attributes*, *+Parser*).

**end**

A close-tag has been parsed. The named handler is called with two arguments: *Handler(+Tag, +Parser)*.

**CDATA**

CDATA has been parsed. The named handler is called with two arguments: *Handler(+CDATA, +Parser)*, where CDATA is an atom representing the data.

**entity**

An entity that cannot be represented as CDATA has been parsed. The named handler is called with two arguments: *Handler(+NameOrCode, +Parser)*.

**pi**

A processing instruction has been parsed. The named handler is called with two arguments: *Handler(+Text, +Parser)*, where *Text* is the text of the processing instruction.

**decl**

A declaration (`<!. . .>`) has been read. The named handler is called with two arguments: *Handler(+Text, +Parser)*, where *Text* is the text of the declaration with comments removed.

This option is especially useful for highlighting declarations and comments in editor support, where the location of the declaration is extracted using `get_sgml_parser/2`.

**error**

An error has been encountered. the named handler is called with three arguments: *Handler(+Severity, +Message, +Parser)*, where *Severity* is one of `warning` or `error` and *Message* is an atom representing the diagnostic message.

The location of the error can be determined using `get_sgml_parser/2`

If this option is present, errors and warnings are not reported using `print_message/3`

**xmlns**

When parsing an in `xmlns` mode, a new namespace declaration is pushed on the environment. The named handler is called with three arguments: *Handler(+NameSpace, +URL, +Parser)*. See section 3.3.1 for details.

**urlns**

When parsing an in `xmlns` mode, this predicate can be used to map a url into either a canonical URL for this namespace or another internal identifier. See section 3.3.1 for details.

### 3.6.1 Partial Parsing

In some cases, part of a document needs to be parsed. One option is to use `load_structure/2` or one of its variations and extract the desired elements from the returned structure. This is a clean solution, especially on small and medium-sized documents. It however is unsuitable for parsing really big documents. Such documents can only be handled with the call-back output interface realised by the `call(Event, Action)` option of `sgml_parse/2`. Event-driven processing is not very natural in Prolog.

The SGML2PL library allows for a mixed approach. Consider the case where we want

to process all descriptions from RDF elements in a document. The code below calls `process_rdf_description(Element)` on each element that is directly inside an RDF element.

```
:- dynamic
    in_rdf/0.

load_rdf(File) :-
    retractall(in_rdf),
    open(File, read, In),
    new_sgml_parser(Parser, []),
    set_sgml_parser(Parser, file(File)),
    set_sgml_parser(Parser, dialect(xml)),
    sgml_parse(Parser,
        [ source(In),
          call(begin, on_begin),
          call(end, on_end)
        ]),
    close(In).

on_end('RDF', _) :-
    retractall(in_rdf).

on_begin('RDF', _, _) :-
    assert(in_rdf).

on_begin(Tag, Attr, Parser) :-
    in_rdf, !,
    sgml_parse(Parser,
        [ document(Content),
          parse(content)
        ]),
    process_rdf_description(element(Tag, Attr, Content)).
```

## 4 Processing Indexed Files

In some cases applications wish to process small portions of large SGML, XML or RDF files. For example, the *OpenDirectory* project by Netscape has produced a 90MB RDF file representing the main index. The parser described here can process this document as a unit, but loading takes 85 seconds on a Pentium-II 450 and the resulting term requires about 70MB global stack. One option is to process the entire document and output it as a Prolog fact-base of RDF triplets, but in many cases this is undesirable. Another example is a large SGML file containing online documentation. The application normally wishes to provide only small portions at a time to the user. Loading the entire document into memory is then undesirable.

Using the `parse(element)` option, we open a file, seek (using `seek/4`) to the position of the element and read the desired element.

The index can be built using the call-back interface of `sgml_parse/2`. For example, the following code makes an index of the `structure.rdf` file of the OpenDirectory project:

```
:- dynamic
    location/3.                                % Id, File, Offset

rdf_index(File) :-
    retractall(location(_, _)),
    open(File, read, In, [type(binary)]),
    new_sgml_parser(Parser, []),
    set_sgml_parser(Parser, file(File)),
    set_sgml_parser(Parser, dialect(xml)),
    sgml_parse(Parser,
        [ source(In),
          call(begin, index_on_begin)
        ]),
    close(In).

index_on_begin(_Element, Attributes, Parser) :-
    memberchk('r:id'=Id, Attributes),
    get_sgml_parser(Parser, charpos(Offset)),
    get_sgml_parser(Parser, file(File)),
    assert(location(Id, File, Offset)).
```

The following code extracts the RDF element with required id:

```
rdf_element(Id, Term) :-
    location(Id, File, Offset),
    load_structure(File, Term,
        [ dialect(xml),
          offset(Offset),
          parse(element)
        ]).
```

## 5 External entities

While processing an SGML document the document may refer to external data. This occurs in three places: external parameter entities, normal external entities and the `DOCTYPE` declaration. The current version of this tool deals rather primitively with external data. External entities can only be loaded from a file and the mapping between the entity names and the file is done using a *catalog* file in a format compatible with that used by James Clark's SP Parser, based on the SGML Open (now OASIS) specification.

Catalog files can be specified using two primitives: the predicate `sgml_register_catalog_file/2` or the environment variable `SGML_CATALOG_FILES` (compatible with the SP package).

### **sgml\_register\_catalog\_file**(*+File, +Location*)

Register the indicated *File* as a catalog file. *Location* is either **start** or **end** and defines whether the catalog is considered first or last. This predicate has no effect if *File* is already part of the catalog.

If no files are registered using this predicate, the first query on the catalog examines **SGML\_CATALOG\_FILES** and fills the catalog with all files in this path.

Two types of lines are used by this package.

```
DOCTYPE doctype file
PUBLIC  " Id " file
```

The specified *file* path is taken relative to the location of the catalog file. For the DOCTYPE declaration, **sgml2pl** first makes an attempt to resolve the **SYSTEM** or **PUBLIC** identifier. If this fails it tries to resolve the *doctype* using the provided catalog files.

Strictly speaking, **sgml2pl** breaks the rules for XML, where system identifiers must be Universal Resource Indicators, not local file names. Simple uses of relative URIs will work correctly under UNIX and Windows.

In the future we will design a call-back mechanism for locating and processing external entities, so Prolog-based file-location and Prolog resources can be used to store external entities.

## 6 Writing markup

The **sgml2pl** package is a parser. Output is generally much easier achieved directly from Prolog. Nevertheless, it contains a few building blocks for emitting markup data.

### **xml\_quote\_attribute**(*+In, -Quoted*)

Map the characters that may not appear in XML attributes to entities. Currently these are `<>&'"`. Future versions may also deal with characters that cannot be represented by the document character-set.

### **xml\_quote\_cdata**(*+In, -Quoted*)

Very similar to **xml\_quote\_attribute/2**, but does not quote the single- and double-quotes.

### **xml\_name**(*+In*)

Succeed if *In* is an atom or string that satisfies the rules for a valid XML element or attribute name.

## 7 Unsupported features

The current parser is rather limited. While it is able to deal with many serious documents, it omits several less-used features of SGML and XML. Known missing SGML features include

- *NOTATION on entities*  
Though notation is parsed, notation attributes on external entity declarations are not handed to the user.
- *NOTATION attributes*  
SGML notations may have attributes, declared using `<!ATTLIST #NOTATION name attributes>`. Those data attributes are provided when you declare an external CDATA, NDATA, or SDATA entity.  
XML does not include external CDATA, NDATA, or SDATA entities, nor any of the other uses to which data attributes are put in SGML, so it doesn't include data attributes for notations either.  
Sgml2pl does not support this feature and is unlikely to; you should be aware that SGML documents using this feature cannot be converted faithfully to XML.
- *SHORTTAG*  
The SGML SHORTTAG syntax is only partially implemented. Currently, `<tag/content/` is a valid abbreviation for `<tag>content</tag>`, which can also be written as `<tag>content</>`. Empty start tags (`<>`), unclosed start tags (`<a<b`) and unclosed end tags (`</a<b`) are not supported.
- *SGML declaration*  
The 'SGML declaration' is fixed, though most of the parameters are handled through indirections in the implementation.
- *The DATATAG feature*  
It is regarded as superseded by SHORTREF, which is supported. (SP does not support it either.)
- *The RANK feature*  
It is regarded as obsolete.
- *The LINK feature*  
It is regarded as too complicated.
- *The CONCUR feature*  
Concurrent markup allows a document to be tagged according to more than one DTD at the same time. It is not supported.

In XML mode the parser recognises SGML constructs that are not allowed in XML. Also various extensions of XML over SGML are not yet realised. In particular, XInclude is not implemented because the designers of XInclude can't make up their minds whether to base it on elements or attributes yet, let alone details.

## 8 Installation

### 8.1 Unix systems

Installation on Unix system uses the commonly found `configure`, `make` and `make install` sequence. SWI-Prolog should be installed before building this package. If SWI-Prolog is not

installed as **pl**, the environment variable **PL** must be set to the name of the SWI-Prolog executable. Installation is now accomplished using:

```
% ./configure
% make
% make install
```

This installs the foreign libraries in **\$PLBASE/lib/\$PLARCH** and the Prolog library files in **\$PLBASE/library**, where **\$PLBASE** refers to the SWI-Prolog ‘home-directory’.

## 9 Acknowledgements

The Prolog representation for parsed documents is based on the SWI-Prolog interface to SP by Anjo Anjewierden.

Richard O’Keefe has put a lot of effort testing and providing bug reports consisting of an illustrative example and explanation of the standard. He also made many suggestions for improving this document.

## A Summary of Predicates

<b>dtd/2</b>	Find or build a DTD for a document type
<b>dtd_property/2</b>	Query elements, entities and attributes in a DTD
<b>free_dtd/1</b>	Free a DTD object
<b>free_sgml_parser/1</b>	Destroy a parser
<b>get_sgml_parser/2</b>	Get parser options
<b>load_dtd/2</b>	Read DTD information from a file
<b>load_html_file/2</b>	Parse HTML file into Prolog term
<b>load_sgml_file/2</b>	Parse SGML file into Prolog term
<b>load_structure/3</b>	Parse XML/SGML/HTML data into Prolog term
<b>load_xml_file/2</b>	Parse XML file into Prolog term
<b>new_dtd/2</b>	Create a DTD object
<b>new_sgml_parser/2</b>	Create a new parser
<b>open_dtd/3</b>	Open a DTD object as an output stream
<b>set_sgml_parser/2</b>	Set parser options (dialect, source, <i>etc.</i> )
<b>sgml_parse/2</b>	Parse the input
<b>sgml_register_catalog_file/2</b>	Register a catalog file
<b>xml_name/1</b>	Test atom for valid XML name
<b>xml_quote_attribute/2</b>	Quote text for use as an attribute
<b>xml_quote_cdata/2</b>	Quote text for use as PCDATA

## Index

#pcdata, 10  
any, 10  
cdata, 9  
content, 15  
declaration, 15  
element, 15  
empty, 9  
explicit, 13  
file, 15  
input, 15  
omitted, 13  
rcdata, 10  
sgml, 12  
shortref, 13  
shorttag, 13  
xmlns, 12  
xml, 12  
-, 6  
/>, 5  
/, 5  
:, 7  
CDATA, 3, 4, 6, 11  
DOCTYPE, 18, 19  
NAMES, 4  
NDATA, 5  
NOTATION, 10  
NUMBERS, 4, 5  
NUMBER, 4, 5  
PUBLIC, 19  
SDATA, 5  
SYSTEM, 19  
#IMPLIED, 11  
#pcdata, 14  
amp, 7  
apos, 7  
begin, 13  
cdata, 10  
conref, 10  
content, 15  
current, 10  
default, 6  
element, 15  
end, 13, 19  
entity, 10  
error, 16  
false, 6, 9  
gt, 7  
idref, 10  
id, 10  
implied, 10  
informational, 15  
integer, 5, 12  
iso-8859-1, 13  
lt, 7  
name, 10  
nmtoken, 10  
notation, 10  
number, 10  
nutoken, 10  
on\_begin, 13, 15  
parse, 13  
preserve, 6, 7  
quot, 7  
rdf-syntax, 8  
rdf, 8  
remove, 7  
required, 10  
sgml, 5-7, 13  
start, 19  
token, 5, 12  
true, 9, 12  
urlns, 8  
utf-8, 13  
warning, 16  
xmlns, 5, 7, 8, 12, 13, 16  
xml, 5, 7, 13  
catalog, 18  
declared content, 9  
dialect, 7  
doctype, 9, 14  
dtd/2, 9  
dtd\_property/2, 9  
event class, 13  
free\_dtd/1, 8  
free\_sgml\_parser/1, 12

get\_sgml\_parser/2, 13

implicit, 11

load\_dtd/2, 8

load\_html\_file/2, 4

load\_sgml\_file/2, 3

load\_structure/3, 4

load\_xml\_file/2, 3

new\_dtd/2, 8

new\_sgml\_parser/2, 11

open\_dtd/3, 9

set\_sgml\_parser/2, 12

sgml\_parse/2, 14

sgml\_register\_catalog\_file/2, 19

shortref, 13

shorttag, 13

xml2pl, 2

xml\_name/1, 19

xml\_quote\_attribute/2, 19

xml\_quote\_cdata/2, 19